koszul-complexes

February 24, 2025

1 Koszul complexes and Cech complexes

In this notebook, we define exterior power, Koszul complexes, and Cech complexes, and give a few examples.

1.1 Exterior powers of a free module

Recall that if F is a free module generated by $e_1, \ldots, e_m$, then $\Lambda^p F$ is the free module generated by $e_{i_1} \wedge \ldots \wedge e_{i_p}$, for all $1 \leq i_1 < i_2 < \cdots < i_p \leq m$. These symbols satisfy $e_i \wedge e_j = -e_j \wedge e_i$, and $e_i \wedge e_i = 0$.

If F is graded free, then so is $\Lambda^p F$.

If π is a permutation on $\{1, 2, \ldots, p\}$, then

$$e_{i_{\pi(1)}} \wedge \ldots \wedge e_{i_{\pi(p)}} = (-1)^{\operatorname{sign} \pi} e_{i_1} \wedge \ldots \wedge e_{i_p}$$

[ ]:
```
R = QQ[a..d];
F = R^{-1} ++ R^{-10} ++ R^{-100} ++ R^{-1000}
```

We have chosen a graded free module with specific degrees designed to be able to see what order Macaulay2 places the basic vectors in.

[ ]:
```
exteriorPower(2, F)
```

Note that the second exterior power has rank 6, and Macaulay2 places the generators in the order

$$e_1 \wedge e_2,\ e_1 \wedge e_3,\ e_2 \wedge e_3,\ e_1 \wedge e_4,\ e_2 \wedge e_4,\ e_3 \wedge e_4.$$

While we are talking about exterior powers, it is worth noting that we can take the exterior power of any R-module M. Think about how you would define this! If

$$0 \leftarrow M \leftarrow F \leftarrow G$$

is a presentation of M, then a presentation of $\Lambda^\ell M$ is given by

$$0 \leftarrow \Lambda^\ell M \leftarrow \Lambda^\ell F \leftarrow \Lambda^{\ell-1} F \otimes_R G.$$

Exercise. Make a definition of $\Lambda^\ell M$, and find a presentation for this (i.e. find the rightmost map in the above sequence).

We will use this construction later, when we look at differential p-forms.

1.1.1 Example: an exterior power

Here we give a simple example of the second exterior power of a module.

[ ]:
```
R1 = QQ[a..f];
```

[ ]:
```
m = matrix{{a,b,c},{b,d,e},{c,e,f}}
```

[ ]:
```
C = res ideal minors(2, m)
```

[ ]:
```
betti C
```

[ ]:
```
M = coker C.dd_3
```

[ ]:
```
exteriorPower(2, M)
```

[ ]:
```
betti res oo
```

1.2 Koszul complex

Let $f_1, \dots, f_m$ be elements of the ring R. Let $F = R^m$. If R and the f_i are graded/homogeneous, then we let F be a graded free module: $F = \bigoplus_{i=1}^m R(-\deg f_i)$.

The **Koszul complex** $K(f_1, \dots, f_m)$ is the R-complex

$$0 \leftarrow R \leftarrow F \leftarrow \Lambda^2 F \leftarrow \cdots \leftarrow \Lambda^m F \leftarrow 0,$$

where the map $\Lambda^p F \to \Lambda^{p-1}$ is defined by

$$e_{i_1} \wedge \dots \wedge e_{i_p} \mapsto \sum_{j=1}^{p} (-1)^{j+1} f_{i_j}\ e_{i_1} \wedge \dots \wedge \widehat{e_{i_j}} \wedge \dots \wedge e_{i_p}.$$

Exercises.

1. Write this out for $m = 1, 2, 3$.
2. Show that $K(f_1, \dots, f_m)$ is a complex.

[ ]:
```
needsPackage "Complexes"
m = matrix{{a,b,c}}
```

[ ]:
```
K = koszulComplex m
```

[ ]:
```
dd^K
```

Here are some important properties of Koszul complexes.

Recall that $f_1, \dots, f_m \in R$ form a a **regular R-sequence** if - $(f_1, \dots, f_m) \neq R$, and - The R-map $f_1 : R \to R$ is injective (i.e. f_1 is a non-zero divisor). - The R-map $f_2 : R/\langle f_1 \rangle \to R/\langle f_1 \rangle$ is injective, - … - The R-map $f_m : R/\langle f_1, \dots, f_{m-1} \rangle \to R/\langle f_1, \dots, f_{m-1} \rangle$ is injective.

Two key theorems involving Koszul complexes and regular sequences are the following.

Theorem A. If $f_1, \dots, f_m$ is a regular R-sequence, then

$$H_p(K(f_1, \dots, f_m)) = \begin{cases} R/\langle f_1, \dots, f_m \rangle & p = 0 \\ 0 & p \geq 1 \end{cases}$$

Consequently, if $f_1, \dots, f_m$ is a regular R-sequence, then the Koszul complex is a free resolution of $R/\langle f_1, \dots, f_m \rangle$.

Theorem B. If R is $\mathbb{N}$-graded, and each f_i is graded of positive degree, then the following are equivalent: 1. $(f_1, \dots, f_m)$ is a regular R-sequence. 2. $H_1(K(f_1, \dots, f_m)) = 0$ 3. $H_p(K(f_1, \dots, f_m)) = 0$, for all $p \geq 1$.

One defines a homogeneous ideal of S to be a **complete intersection** if its codimension (computed e.g. via Hilbert series, polynomials) is equal to the number of generators. With this definition, $f_1, \dots f_m$ is a regular R-sequence if and only if $\langle f_1, \dots, f_m \rangle$ is a complete intersection.

Remarkably, if the Koszul complex is exact at spot 1, it is exact at all spots after that.

1.2.1 What it the ideal is not a complete intersection?

[ ]:
```
R = QQ[a..f];
I = ideal(a^2*b-c^2*d, a*b*c-d*e*f, a*d^2-b*f^2)
```

[ ]:
```
C = koszulComplex I_*
```

[ ]:
```
dd^C
```

[ ]:
```
prune HH_1(C)
```

[ ]:
```
prune HH_2(C)
```

[ ]:
```
prune HH_3(C)
```

Exploratory question: What can you say about the higher $H_i(K(f_1, \dots, f_m))$ in case $(f_1, \dots, f_m)$ is not a regular sequence? For instance, when are they non-zero?

Here are a few key facts about Koszul complexes. Let $K^{(j)} := K(f_j, \dots, f_m)$. Suppose that $\langle f_1, \dots, f_m \rangle \neq R$.

1. For any $f_1, \dots, f_m \in R$, there is an exact sequence of R-complexes

   $$0 \longleftarrow K^{(2)}[-1] \longleftarrow K^{(1)} \longleftarrow K^{(2)} \longleftarrow 0.$$

   (and so consequently, there is a long exact sequence in homology, which is multiplication by $f_1 \cdot$).
2. $H^i(K^{(m)}) = 0$ for all $i > 0$ if and only if $(f_1, \dots, f_m)$ is a regular sequence.

Let's illustrate these results using the `Complexes` package.

[ ]:
```
K2 = koszulComplex{b,c}; dd^K2
```

[ ]:
```
K3 = koszulComplex{a,b,c}; dd^K3
```

```
[ ]: f = map(K2, K2 ** R^{-1}, a * id_K2)
```

```
[ ]: (degree f, source f == K2 ** R^{-1}, target f == K2, isHomogeneous f)
```

```
[ ]: E = cone f
```

```
[ ]: dd^E
```

```
[ ]: E == K3
```

```
[ ]: F = canonicalMap(E, target f)
```

```
[ ]: G = canonicalMap((source f)[-1], E)
```

```
[ ]: isShortExactSequence(G, F)
```

```
[ ]: connectingMap(G, F)
```

```
[ ]: prune HH E
```

```
[ ]: prune HH K2
```

1.3 The Cech complex

Let $S = k[x_0, \dots, x_n]$ be the homogeneous coordinate ring of $\mathbb{P}^n$.
Define the **Cech complex** to be the (graded) S-complex $\mathscr{C}(S)$

$$0 \to \mathscr{C}^0 \to \mathscr{C}^1 \to \cdots \to \mathscr{C}^n \to 0$$

where

$$\mathscr{C}^p := \bigoplus_{i_0 < i_1 < \cdots < i_p} S[x_{i_0}^{-1}, \dots, x_{i_p}^{-1}]$$

and the differential

$$\partial^p : \mathscr{C}^p \to \mathscr{C}^{p+1}$$

is given by, if $\omega = (\omega_\alpha)_\alpha$ where $\alpha = \{i_0, i_1, \dots, i_p\}$, then

$$\partial^p(\omega)_{\{i_0,\dots,i_{p+1}\}} := \sum_{j=0}^{p+1} (-1)^j \omega_{\{i_0,\dots,\widehat{i_j},\dots,i_{p+1}\}}$$

1.4 Definition of sheaf cohomology and first results of Serre

If M is a graded S-module, then $M \otimes_S \mathscr{C}$ is a graded complex of (infinitely generated) graded S-modules. Let's denote by $(M \otimes_S \mathscr{C})_d$ the degree d part of this complex.

Definition Let M be a graded S-module, and let $\widetilde{M}$ be the corresponding coherent sheaf on $\mathbb{P}^n$.
The sheaf cohomology of $\widetilde{M}$ is

$$H^i(\widetilde{M}) := H^i(M \otimes_S \mathscr{C})_0.$$

1.5 The graded $\Bbbk$-dual of a module

1.5.1 An example

Let's start with an example.

[ ]:
```
S = ZZ/32003[a..d];
```

[ ]:
```
I = ideal(a^2, b^2, c^2, d^3, a*b, c*d, a^2*d)
```

[ ]:
```
M = S^1/I
```

[ ]:
```
for d  from -1 to 4 list hilbertFunction(d, M)
```

[ ]:
```
Mdual = Ext^4(M, S^{-4})
```

[ ]:
```
for d  from -4 to 1 list hilbertFunction(d, Mdual)
```

Cool! This is kind of like M "upside-down"…

It turns out we can do this for *any* graded S-module.

Definition. Given a $\mathbb{Z}$-graded S-module M, define the $\mathbb{Z}$-graded S-module $\check{M}$ to be

$$\check{M} \;=\; \bigoplus_{d \in \mathbb{Z}} \check{M}_d,$$

where $\check{M}_d := (M_{-d})^*$ (the $\Bbbk$-dual vector space).

Multiplication is induced by the natural map

$$M^*_{-d} \otimes S_1 \to M^*_{-d-1}.$$

Exercises. 1. Write the definition out carefully, and check that this gives a well-define graded S-module. 2. Show that the graded $\Bbbk$-dual of $\check{M}$ is again M. 3. Show: M is finitely generated, if and only if $\check{M}$ is zero in all high enough degrees (i.e. the corresponding sheaf is zero).

Given that the dual of a finitely generated M is only finitely generated if M is finite dimension over the base field (also called *Artinian*), we tend to only actually compute this if M has finite dimension over the base field.

Exercise. 4. If M is a graded Artinian finitely-generated S-module, show that

$$\check{M} = \operatorname{Ext}^{n+1}_S(M, S(-n-1)).$$

(hint: consider a free resolution of M, and recall that the transpose gives a resolution of $\operatorname{Ext}^{n+1}(M, S)$. Now compute Hilbert series)

1.5.2 Example: the graded $\Bbbk$-dual to S

Exercise 5. Compute the graded $\Bbbk$-dual of S. Show that it can be represented as $\check{S} = \Bbbk[\frac{1}{x_0}, \frac{1}{x_1}, \dots, \frac{1}{x_n}]$, and give the S-module structure on $\check{S}$.

1.6 Serre's FAC paper results about computing cohomology of sheaves.

[ ]:
```

```

[ ]:
```

```