

hodge-diamonds

February 24, 2025

1 The Hodge diamond of a smooth projective variety

Let $S = \mathbb{k}[x_0, \dots, x_n]$ be the homogeneous coordinate ring of \mathbb{P}^n . Let $X \subset \mathbb{P}^n$ be a smooth projective variety, with homogeneous ideal $I \subset S$.

In this notebook, we discuss 1. Computing the sheaf Ω_X^p of p -forms on X 2. Define the Hodge diamond of X , and some basic theorems about it. 3. Try to find fast ways to compute this!

While you read this, try to apply the methods (in Macaulay2) to (some of) the following cases:

1. A surface given by a degree d polynomial in \mathbb{P}^3 , for various d .
2. A 3-fold given by a degree d polynomial in \mathbb{P}^4 , for various d .
3. A variety given by the 3×3 minors of a 3×4 matrix of random linear forms.

1.1 The sheaf of 1-forms on X

A graded S -module representing the sheaf of 1-forms on a projective variety X (with ideal $I \subset S$) can be computed using the following two exact sequences.

First we have the so-called Euler sequence:

$$0 \longrightarrow \Omega_{\mathbb{P}^n}^1 \longrightarrow \mathcal{O}_{\mathbb{P}^n}(-1)^{n+1} \longrightarrow \mathcal{O}_{\mathbb{P}^n} \longrightarrow 0$$

In other words the sheaf of 1-forms on \mathbb{P}^n is the sheaf associated to the graded module which is the kernel of the map $S(-1)^{n+1} \rightarrow S$ given by the variables of S .

Second, one can compute the sheaf of 1-forms of a subvariety $X \subset \mathbb{P}^n$ with ideal $I \subset S$ using the following right exact sequence of \mathcal{O}_X modules.

$$I/I^2 \longrightarrow \Omega_{\mathbb{P}^n}^1 \otimes \mathcal{O}_X \longrightarrow \Omega_X^1 \longrightarrow 0$$

The first map sends a polynomial $f \in I$ to $df \otimes 1$.

Note: This sequence is exact on the left exactly when X is a smooth variety.

Computing this module

Suppose that $I = \langle f_1, \dots, f_r \rangle \subset S$ is the ideal of a projective variety $X \subset \mathbb{P}^n$. Let $F = \bigoplus_{i=1}^r S(-\deg f_i)$, and let $\varphi : F \rightarrow I \subset S$ be the matrix sending the i -th basis element to f_i .

Let $R = S/I$ be the quotient ring. We have two R -maps of interest: 1. The Koszul map $d : R(-1)^{n+1} \rightarrow R$ which maps the i -th basis element to the image of x_i in R . 2. The Jacobian map $\text{jac} : F \otimes_S R \rightarrow R(-1)^{n+1}$ with matrix $(\partial f_i / \partial x_j)$.

Exercises 1. Show that $d \circ \text{jac} = 0$. 2. Show that $M := \ker(d) / \text{image}(\text{jac})$ satisfies $\widetilde{M} \cong \Omega_X^1$. 3. This M is a graded R -module, and hence also a graded S -module. Show how to compute an S -presentation of M . 4. Find this presentation for some of the example varieties mentioned above.

1.1.1 Example: sheaf of 1-forms of a hypersurface

Let's consider hypersurfaces in \mathbb{P}^3 . Let's get a feel for what the module constructed here looks like, and how we might improve it.

```
[ ]: S = ZZ/101[x_0..x_3];
[ ]: I = ideal random(2, S)
[ ]: jac = jacobian I
[ ]: vars S
[ ]: R = S/I
[ ]: ker (vars R)
[ ]: jacobian I % oo
[ ]: f = modulo((syz vars S)**R, (jacobian I)**R)
[ ]: fS = lift(f, S)
[ ]: M = coker fS
[ ]: ann M == I
[ ]: betti res M
[ ]: S = QQ[x_0..x_4];
[ ]: I = ideal(x_0^5 + x_1^5 + x_2^5 + x_3^5 + x_4^5 - x_0*x_1*x_2*x_3*x_4);
[ ]: R = S/I;
[ ]: d = (vars S) ** R
[ ]: jac = (jacobian I) ** R
[ ]: d * jac
```

```
[ ]: m = presentation ((kernel d)/(image jac))
[ ]: image((syz vars S)**R) + image jac == kernel d
[ ]: mS = lift(m, S);
    M = prune coker mS
[ ]: ann M == I
[ ]: betti res M
```

Here are some other things one can look at. From this, what is the cohomology of the sheaf Ω_X^1 ?

```
Ext^0(M, S^{-5})
E1 = Ext^1(M, S^{-5})
E2 = Ext^2(M, S^{-5})
E3 = Ext^3(M, S^{-5})
hilbertFunction(0, E1)
hilbertFunction(0, E2)
hilbertFunction(0, E3)
```

```
[ ]: mS = lift(m, S)
[ ]: betti res coker mS
[ ]: numgens S
[ ]: C = res coker mS
[ ]: Mpurified = HH^0((sheaf coker mS)(>=0))
[ ]: betti res M
[ ]: betti res Mpurified -- same as M, not always the case!
```

Let's check to see if X is smooth.

```
[ ]: singX = ideal jacobian I
[ ]: R = S/I
[ ]: codim singX == 5 -- so it is smooth
```

1.2 The sheaf Ω_X^p of p -forms on X

Since $\Omega_X^p = \Lambda^p \Omega_X^1$, once we have a module M which represents Ω_X^1 , we need only compute $\Lambda^p M$.

1.2.1 Example: sheaf of p -forms of a hypersurface

```
[ ]: M2 = exteriorPower(2, M);
      betti res M2

[ ]: M2purified = HH^0((sheaf M2)(>=0))

[ ]: betti res M2purified

[ ]: HH^1(sheaf M2)

[ ]: hilbertFunction(0, Ext^3(M2, S^{-5}))

[ ]: hilbertFunction(0, Ext^2(M2, S^{-5}))
```

1.3 Hodge diamond

The Hodge diamond of a smooth projective variety X of dimension d is the matrix of numbers $h^{p,q}(X) := h^q(\Omega_X^p)$.

Usually it is written as a diamond. Here we will write it as a matrix of integers...

Writing a Macaulay2 function which displays it as a diamond of numbers would be a good exercise in writing Macaulay2 code!

$$\begin{pmatrix} h^{0,0}(X) & h^{0,1}(X) & \dots & h^{0,d}(X) \\ h^{1,0}(X) & h^{1,1}(X) & \dots & h^{1,d}(X) \\ \vdots & \vdots & & \vdots \\ h^{d,0}(X) & h^{d,1}(X) & \dots & h^{d,d}(X) \end{pmatrix}$$

By definition, $\Omega_X^0 = \mathcal{O}_X$, so the first row of this matrix is the cohomology vector

$$h^\bullet(\mathcal{O}_X) := (h^0(\mathcal{O}_X), h^1(\mathcal{O}_X), \dots, h^d(\mathcal{O}_X)).$$

Similarly, the next rows are $h^\bullet(\Omega_X^1), \dots, h^\bullet(\Omega_X^d)$.

Before examining this more closely, let's compute it for a quartic surface in 3-space, using built-in functionality in Macaulay2

```
[ ]: S = ZZ/32003[x_0..x_3];

[ ]: F = x_0^4 + x_1^4 + x_2^4 + x_3^4 + 5*x_0*x_1*x_2*x_3

[ ]: X = variety ideal F
```

First, we check that this is a non-singular quartic.

```
[ ]: singF = ideal jacobian F

[ ]: saturate singF
```

There are no singular points, so yes, it is smooth.

Alternatively, we could have just computed the codimension. This does a simpler Groebner basis. If the codimension is $n + 1 = 4$, then it is non-singular.

```
[ ]: codim singF == 4
```

```
[ ]: matrix for p from 0 to 2 list for q from 0 to 2 list rank_
      ↪ HH^q(cotangentSheaf(p, X))
```

Let's write down some important facts about these numbers. We suppose here that X is a smooth projective variety of dimension d .

1. Hodge theory: $h^{p,q}(X) = h^{q,p}(X)$.
2. Serre duality: $h^{p,q}(X) = h^{d-p,d-q}(X)$.
3. Hodge theory: $\text{rank } H^r(X, \mathbb{Z}) = \sum_{p+q=r} h^{p,q}(X)$.
4. Basics: $h^{p,q}(X) = 0$ if $p < 0$, $q < 0$, $p > d$, or $q > d$.

Exercises

1. Given a smooth projective surface X , or a smooth projective 3-fold, find a smallest set of cohomologies that one needs to compute in order to know the entire Hodge diamond.
2. Using Euler characteristics, one can get by with even fewer cohomology computations. What might be a good set?

For this exercise, try computing the Hodge diamond “by hand” i.e. by computing a few cohomologies, Euler characteristics, etc, for the following cases.

1. A surface given by a degree d polynomial in \mathbb{P}^3 , for various d .
2. A 3-fold given by a degree d polynomial in \mathbb{P}^4 , for various d .
3. A variety given by the 3×3 minors of a 3×4 matrix of random linear forms.

– Here is an example to get you started.

```
S = ZZ/32003[x_0..x_4];
I = ideal random(5, S);
M0 = S^1/I
--M1 = cotangentModule(1, I) -- defined earlier in this notebook. This gives the module as an
--M2 = cotangentModule(2, I) -- exercise question: do we need this one?
X = variety I
HH^1(OO_X)
HH^2(OO_X)
Ext^1(S^1/I, S^{-5})
hilbertFunction(0, Ext^1(S^1/I, S^{-5}))
basis(0, Ext^1(S^1/I, S^{-5}))
```

1.4 Stop!! Try these exercises before reading on!!

```
[ ]:
```

```
[ ]:
```