

primary-decompositions

March 2, 2025

0.1 Minimal and associated primes, primary ideals, primary decomposition

In order to understand an object (in this case, an ideal, or an algebraic set), it is useful to break it up into “simpler” pieces. For us, that means finding the irreducible components of an algebraic set, and finding the minimal primes, and primary decomposition of an ideal.

0.1.1 Definitions and basic facts

In class, I expect that you have seen these concepts, but here is a reminder and a few exercises to play with!

Definition. Let R be a Noetherian ring, and $I \subseteq R$ a proper ideal. Then

- I is **maximal** if there is no proper ideal containing I other than R .
- I is **prime** if $ab \in I$ implies $a \in I$ or $b \in I$.
- I is **primary** if $ab \in I$ implies $a \in I$ or $b^m \in I$, for some $m \in \mathbb{N}$.
- I is **irreducible** if I cannot be written as $I_1 \cap I_2$ for two ideals with $I_i \neq I$.
- I is **radical** if $f^m \in I$ implies $f \in I$.

Exercise (good one to do if you have not done this before!)

1. Which monomial ideals are maximal? prime? irreducible?
2. Which monomial ideals are primary?

Exercises

1. Every maximal ideal is a prime ideal.
2. I primary implies that $P = \sqrt{I}$ is prime. I is called a P -primary ideal.
3. If the radical P of I is prime, is I a P -primary ideal? (answer: no! Such an ideal is sometimes called a **pseudo-primary** ideal).
4. Show: The intersection of two P -primary ideals is again P -primary.

Exercise/Proposition. Every proper ideal $I \subset R$ in a Noetherian ring is an intersection of a finite number of irreducible ideals.

A **primary decomposition** of an ideal I is a representation

$$I = Q_1 \cap Q_2 \cap \cdots \cap Q_r,$$

where each Q_i is P_i -primary. It is **irredundant** if removing any Q_i results in an intersection which doesn't equal I , and if P_1, \dots, P_r are pairwise distinct. (I often say: $I = Q_1 \cap Q_2 \cap \cdots \cap Q_r$ is a **PD** to mean that it is an *irredundant* primary decomposition).

```
[ ]: --%timeout=40000
printWidth=0
```

0.1.2 Example: Monomial ideals

Exercise. By hand, find an irreducible decomposition of the following monomial ideal. Later we show that each irreducible ideal is primary. Using this, find a *PD* (irredundant primary decomposition) of the following monomial ideal. Check your work using Macaulay2.

```
[ ]: R = QQ[a..d];
M = monomialIdeal(a^2, a*b, b^3*c, c^4);
```

```
[ ]: isPrime M
```

```
[ ]: isPrimary M
```

```
[ ]: radical M
```

```
[ ]: primaryDecomposition M
```

```
[ ]: irreducibleDecomposition M
```

0.2 Ideal quotients and saturations

Recall that $(I : g) := \{f \in R \mid fg \in I\}$.

Note that $I \subseteq (I : g) \subseteq (I : g^2) \subseteq \dots$ is an increasing set of ideals, and since R is Noetherian, this eventually stabilizes. We let $(I : g^\infty) := \bigcup_{m=0}^{\infty} (I : g^m)$ denote this ideal. It is called the **saturation of I with respect to g** . Note that there is always a minimal $\ell \geq 0$ such that $(I : g^\infty) = (I : g^\ell)$.

In Macaulay2, this is denoted as `saturate(I, g)`. Earlier, we have seen ways to compute both $(I : g)$ and $(I : g^\infty)$ using Groebner bases.

Exercises.

- Show that if I is irreducible, then I is primary.
- Deduce from this and stuff above that every proper ideal $I \subset R$ in a Noetherian ring has an irredundant primary decomposition.

0.2.1 What about uniqueness? Associated and minimal primes.

Here is an important pair of lemmas about the relationship of primary ideals and primary decompositions, and ideal quotients and saturations. I will leave these as (good!) exercises for you to do.

Let $Q \subset R$ be a P -primary ideal. Let $f \in R$.

Proposition/Exercise.

$$Q : f^\infty = \begin{cases} \langle 1 \rangle & \text{if } f \in P \\ Q & \text{if } f \notin P \end{cases}$$

Proposition/Exercise.

$$Q : f = \begin{cases} \langle 1 \rangle & \text{if } f \in Q \\ P\text{-primary} & \text{if } f \in P, \text{ but } f \notin Q \\ Q & \text{if } f \notin P. \end{cases}$$

Suppose we are given two (irredundant) primary decompositions:

$$I = Q_1 \cap \cdots \cap Q_r = Q'_1 \cap \cdots \cap Q'_s,$$

where Q_i is P_i -primary, and Q'_i is P'_i -primary.

What can we say about the r, s ? do the same primes occur? do the same primaries occur?

Definition. Let R be a Noetherian ring, and $I \subseteq R$ a proper ideal. Then

- P is a **minimal prime** of I if $I \subseteq P$, and if $I \subseteq P' \subseteq P$ for another prime ideal P' , then $P' = P$.
- A prime ideal P is an **associated prime** of I if there exists $f \in R$ such that $P = (I : f)$.

Exercise. Show that the set of associated primes of I is $\{P_1, \dots, P_r\}$. Consequently, $r = s$, and $\{P_1, \dots, P_r\} = \{P'_1, \dots, P'_s\}$.

Exercise. Show that the minimal (under inclusion) elements of the set of associated primes is the set of minimal primes defined above.

Each associated prime is either **minimal** or **embedded** (that is, not minimal).

Exercise Use these results to determine what aspects of a PD are unique. Be mindful of the following example!

$$\langle x^2, xy \rangle = \langle x \rangle \cap \langle x^2, y \rangle = \langle x \rangle \cap \langle x^2, xy, y^m \rangle, \text{ for any } m!$$

For instance, given a PD $I = Q_1 \cap \cdots \cap Q_m$, where Q_i is P_i -primary. are the following unique?

- The set of primes P from a PD .
- The minimal set of primes from a PD .
- The primary ideal Q corresponding to a minimal element of this set of primes.
- The primary ideal Q corresponding to an embedded prime (an *embedded prime* is any P_i which is not minimal).

0.2.2 Example: Simplicial complexes and square-free monomial ideals

Definition. An (abstract) *simplicial complex* Δ on a vertex set V is a set of subsets of V such that

- For each $i \in V$, $\{i\} \in \Delta$.
- If $\tau \subseteq \sigma \in \Delta$, then $\tau \in \Delta$.

Here is some terminology about simplicial complexes.

- Every element $\sigma \in \Delta$ is called a **face** of Δ .
- Every subset $\tau \notin \Delta$ is called a **nonface** of Δ .
- Just to be confusing: a **coface** of Δ is a subset whose complement in V is a face.

For example, the boundary of the tetrahedron with vertices labeled 1, 2, 3, 4 is the simplicial complex Δ with maximal faces in the set $\{\{1, 2, 3\}, \{1, 2, 4\}, \{1, 3, 4\}, \{2, 3, 4\}\}$. Note that since any subset of a face is also a face, the maximal faces completely determine Δ .

Note that if I choose a set of subsets of V , then this determines uniquely a simplicial complex: all subsets of all of these sets. That is, a simplicial complex is uniquely determined by giving its maximal faces.

The Stanley-Reisner ideal

Time to relate this to commutative algebra!

Definition. Given a simplicial complex Δ on a vertex set (say) $\{1, 2, \dots, n\}$, then define the Stanley-Reisner ideal $I_\Delta \subset \mathbb{k}[x_1, \dots, x_n]$ to be the ideal generated by monomials $m_\tau := \prod_{i \in \tau} x_i$ for each nonface τ of Δ .

For example, if Δ is the simplicial complex with maximal faces $\{1, 2, 3\}, \{1, 4\}, \{2, 4\}, \{3, 4\}$, whose nonfaces are $\{1, 2, 4\}, \{1, 3, 4\}, \{2, 3, 4\}$, and $\{1, 2, 3, 4\}$, then $I_\Delta = \langle x_1 x_2 x_4, x_1 x_3 x_4, x_2 x_3 x_4 \rangle$. I_Δ is minimally generated by monomials corresponding to the minimal nonfaces of Δ .

Exercise. Show that *any* squarefree monomial ideal (i.e. an ideal generated by a set of squarefree monomials) is I_Δ for a uniquely given simplicial complex.

Our goal: finding the primary decomposition of the Stanley-Reisner ideal I_Δ

Note that I_Δ is a radical ideal, and therefore its primary decomposition is given by the intersection of all of the associated primes of I_Δ . (exercise: prove these assertions!)

It will turn out also that each associated prime is a monomial prime ideal.

Important exercise. Given these, choose a couple of smaller simplicial complexes (on e.g. 3,4,5 vertices), and try to compute the primary decomposition of the corresponding I_Δ . Can you find a general pattern?

Stop! Do not read on until you have tried some examples, and tried to find a general pattern!

Here is one example done using Macaulay2.

```
[ ]: R = QQ[x_1..x_4];
      I = ideal(x_1*x_2*x_4, x_1*x_3*x_4, x_2*x_3*x_4)
```

```
[ ]: primaryDecomposition I
```

Stop! Once you have done some examples, read on!

Given Δ , suppose that $\sigma = \{i_1, \dots, i_r\} \in \Delta$ is a maximal face. Therefore $\{i_1, \dots, i_r, k\}$ is a nonface, for every $k \notin \sigma$. Note that then

$$(I_\Delta : x_{i_1} x_{i_2} \dots x_{i_r}) = \langle x_k \mid k \notin \sigma \rangle.$$

Therefore

$$I_\Delta \subseteq \bigcap_{\tau \text{ is a minimal coface of } \Delta} \langle x_k \mid k \in \tau \rangle.$$

Note that if m is a squarefree monomial in the right-hand side of this equation, then m corresponds to a nonface of Δ , and so is contained in I_Δ .

Therefore

$$I_{\Delta} = \bigcap_{\tau \text{ is a minimal coface}} \langle x_k \mid k \in \tau \rangle$$

is a primary decomposition of I_{Δ} !

Example from above: Continuing with the example from above, if Δ is the simplicial complex with maximal faces $\{1, 2, 3\}, \{1, 4\}, \{2, 4\}, \{3, 4\}$, whose nonfaces are $\{1, 2, 4\}, \{1, 3, 4\}, \{2, 3, 4\}$, and $\{1, 2, 3, 4\}$, where $I_{\Delta} = \langle x_1x_2x_4, x_1x_3x_4, x_2x_3x_4 \rangle$, then the minimal cofaces are the complements of the maximal faces: $\{4\}, \{2, 3\}, \{1, 3\}, \{1, 2\}$, and so the primary decomposition is

$$I_{\Delta} = \langle x_1x_2x_4, x_1x_3x_4, x_2x_3x_4 \rangle = \langle x_4 \rangle \cap \langle x_2, x_3 \rangle \cap \langle x_1, x_3 \rangle \cap \langle x_1, x_2 \rangle$$

I_{Δ} is minimally generated by monomials corresponding to the minimal nonfaces of Δ .

0.2.3 A divide and conquer algorithm

Key Lemma. (We did this in class) Let $I \subset R$ be an ideal, and $f \in R$ not be in I . Suppose that $(I : f^{\infty}) = (I : f^{\ell})$. Then

$$I = (I : f^{\infty}) \cap (I + \langle f^{\ell} \rangle).$$

One can use the key lemma to compute a primary decomposition, if one can find elements f which are zero divisors mod I . However, this will often result in a highly redundant primary decomposition.

Here is an example of this. Note that in Macaulay2, `saturate(I, f)` refers to the saturation $(I : f^{\infty})$.

```
[ ]: I = ideal(c^2-b*d, b^2-a*c)
```

Exercise. Using Macaulay2 to do intersections, ideal quotients, saturations, try to find a *PD* of I . Check your work. Try this on your own before looking at the answer below! There are questions here: How to find good polynomials to split by? How to know when you have a prime, or primary ideal?

```
[ ]: compsI = primaryDecomposition I
```

```
[ ]: saturate(I, b) == (I : b)
```

```
[ ]: I1 = I:b
```

```
[ ]: I2 = trim(ideal(b) + I)
```

```
[ ]: intersect(I1, I2) == I
```

Now split I_2 using $f = a$ (Note that using c would not split the ideal!)

```
[ ]: I21 = saturate(I2, a)
```

```
[ ]: I21 == I2 : a
```

```
[ ]: I22 = trim(ideal(a) + I2)
```

```
[ ]: I == intersect(I1, I21, I22)
```

Note that I_{22} is redundant!

```
[ ]: I == intersect(I1, I21)
```

0.3 Special case: Zero-dimensional ideals

We are going to spend some effort understanding what looks to be a special case: zero dimensional ideals, i.e. ideals defining a finite set of points.

It turns out that to compute a primary decomposition of an ideal $I \subset \mathbb{k}[x_1, \dots, x_n]$ we will be able to reduce this to a number of computations in the zero dimensional case.

So, suppose that $I \subset R = \mathbb{k}[x_1, \dots, x_n]$ is a *zero-dimensional* ideal, that is, R/I is a finite dimensional vector space over \mathbb{k} .

Recall that if \bar{k} denotes the algebraic closure of \mathbb{k} , then we let $\mathbb{A}^n := \bar{k}^n$, and define

$$\mathbb{V}(I) := \{p \in \mathbb{A}^n \mid f(p) = 0, \text{ for all } f \in I\}$$

Geometrically, this means that the zero set of I , $X = \mathbb{V}(I) \subset \mathbb{A}^n$ is a set of points. The question: how do we find these points?

Before we give a general algorithm, let's take a look at a non-trivial example. This is the so-called cyclic roots problem. This particular case is zero-dimensional, and has 70 solutions, and is a radical ideal. We will play with it, and see how it goes, and what is involved. The fact that it is a radical ideal will mean we need to try other examples that are not radical, but this still gives a good idea of what is involved, and what directions we might take to find the minimal primes (same as the primary decomposition, as the ideal is radical).

```
[ ]: R = QQ[x_1..x_5]
```

```
[ ]: I = ideal(x_1 + x_2 + x_3 + x_4 + x_5,
  x_1*x_2 + x_2*x_3 + x_3*x_4 + x_4*x_5 + x_5*x_1,
  x_1*x_2*x_3 + x_2*x_3*x_4 + x_3*x_4*x_5 + x_4*x_5*x_1 + x_5*x_1*x_2,
  x_1*x_2*x_3*x_4 + x_2*x_3*x_4*x_5 + x_3*x_4*x_5*x_1 + x_4*x_5*x_1*x_2 +
  ↪x_5*x_1*x_2*x_3,
  x_1*x_2*x_3*x_4*x_5 - 1
)
```

```
[ ]: groebnerBasis I;
```

```
[ ]: codim I, degree I
```

```
[ ]: numcols basis(R^1/I)
```

```
[ ]: eliminate(I, {x_2,x_3,x_4,x_5})
```

```
[ ]: f1 = oo_0
```

```
[ ]: factor f1
[ ]: radical I == I
[ ]: compsI = primaryDecomposition I;
[ ]: #compsI
[ ]: compsI/dim
[ ]: compsI/degree
[ ]: sum oo
[ ]: Rlex = (coefficientRing R)[gens R, MonomialOrder => Lex];
      Ilex = sub(I, Rlex);
      Igb = ideal groebnerBasis Ilex;
      netList Igb_*
[ ]: netList compsI
[ ]: #compsI
[ ]: compsI_0
[ ]: I == intersect compsI
[ ]: compsI/isPrime
```

Let's change coordinates to make this in x_n -general position.

```
[ ]: use Rlex
      phi = map(Rlex, Rlex, {x_1, x_2, x_3, x_4, x_5-2*x_1+5*x_2+7*x_3-3*x_4})
[ ]: I' = phi Ilex;
[ ]: gbI' = ideal groebnerBasis I';
[ ]: netList gbI'_*
[ ]: facs = factor gbI'_0
[ ]: facts = (toList facs)/toList/first
[ ]: --%timeout=40000
      ids = for f in facts list ideal groebnerBasis(I' + ideal f);
[ ]: ids_0
```

```
[ ]: ids1 = for J in ids list ideal groebnerBasis(phi^-1 J);
```

```
[ ]: Ilex == intersect ids1
```

0.3.1 Facts about zero-dimensional ideals

Suppose that I is a radical zero-dimensional ideal.

Exercises

1. (Shape Lemma) If the x_n components of the zeros of I are all distinct, and there are d such points, then the reduced Groebner basis of I with respect to the lexicographic with $x_1 > x_2 > \cdots > x_n$ has the form:

$$\begin{pmatrix} x_1 - f_1(x_n) \\ x_2 - f_2(x_n) \\ \vdots \\ x_{n-1} - f_{n-1}(x_n) \\ f_n(x_n) \end{pmatrix},$$

where the degree of f_n is d , and the degree of f_i is $< d$, for all $i < n$.

```
[ ]:
```